# LuckPay Web Manage System
#
# Copyright (c) 2016 Lucky Byte, Inc.
#
$ ->
    form = $('form.main')

    $('body').css('background-color', 'black')

    # 过滤一致的记录
    form.find('#same_filter').click (evt) ->
        evt.preventDefault()
        if $(this).data('filted')
            form.find('tr[data-same=true]').show('fast')
            $(this).data('filted', false)
        else
            form.find('tr[data-same=true]').hide('fast')
            $(this).data('filted', true)

    # 全选
    form.find('#check_all').click (evt) ->
        evt.preventDefault()
        form.find('td .checkbox').checkbox 'set checked'
        form.find('button#pass').removeClass('disabled')

    # 勾选时检查是否全部复核通过，并更新'通过按钮'的状态
    form.find('td .checkbox').each ->
        $(this).checkbox
            onChecked: ->
                for box in form.find('td .checkbox')
                    if not $(box).checkbox('is checked')
                        form.find('button#pass').addClass('disabled')
                        return
                form.find('button#pass').removeClass('disabled')
            onUnchecked: ->
                form.find('button#pass').addClass('disabled')

    # 取消按钮事件
    form.find('button#cancel').click ->
        window.location = '/terms'

    # 拒绝按钮事件
    form.find('button#reject').click ->
        if $(this).hasClass('has_clicked')
            reason = form.form 'get value', 'reject_reason'
            if not reason or reason.trim().length == 0
                alert '请输入拒绝的原因，以便录入人员可以有针对性的进行修改'
                return
            $(this).removeClass('has_clicked').text('拒 绝')
            form.attr 'action', '/terms/check/reject'
            form.submit()
        else
            form.find('button#pass').text('通 过').removeClass('has_clicked')
            form.find('.fields#effect_time').hide()
            form.find('.fields#reject_reason').show()
            $(this).addClass('has_clicked').text('确认拒绝')

    # 通过按钮事件
    form.find('button#pass').click ->
        if $(this).hasClass('has_clicked')
            effect_time = form.form 'get value', 'effect_time'
            if not effect_time or effect_time.trim().length == 0
                alert '请选择信息生效的时间'
                return
            $(this).removeClass('has_clicked').text('通 过')
            form.attr 'action', '/terms/check/pass'
            form.submit()
        else
            form.find('button#reject').text('拒 绝').removeClass('has_clicked')
            form.find('.fields#reject_reason').hide()
            form.find('.fields#effect_time').show()
            $(this).addClass('has_clicked').text('确认通过')
